Maple-assisted proof of formula for A295270

Robert Israel

17 November 2017

There are $2^6 = 64$ possible configurations for a 2×3 sub-array. Consider the 64×64 transition matrix T such that $T_{ij} = 1$ if the bottom two rows of a 3×3 sub-array could be in configuration i while the top two rows are in configuration j (i.e. the middle row is compatible with both i and j, and each 1 in that row is horizontally or vertically adjacent to 0, 1 or 4 1's), and 0 otherwise. The following Maple code computes it. I'm encoding a configuration

$$\begin{bmatrix} b_1 & b_2 & b_3 \\ b_4 & b_5 & b_6 \end{bmatrix}$$

as $b + 1$ where $b_1 b_2 b_3 b_4 b_5 b_6$ is the binary representation of b. The $+1$ is needed because matrix indices start at 1 rather than 0.

```
> q:= proc(a,b) local r,s,t,M,i;
     s:= floor((a-1)/8);
     if s <> (b-1) mod 8 then return 0 fi;
     s:= convert(s+8,base,2);
     r:= convert(8+floor((b-1)/8),base,2);
     t:= convert(8+ ((a-1) mod 8),base,2);
     M:= Vector(3);
     if s[1] = 1 and s[2] = 1 then M[1]:= 1; M[2]:= 1 fi;
     if s[2]=1 and s[3]=1 then M[2]:= M[2]+1; M[3]:= 1 fi;
     for i from 1 to 3 do if s[i]=1 then
       M[i]:= M[i]+r[i]+t[i];
       if M[i] = 2 or M[i]=3 then return 0 fi;
     fi od;
     1
  end proc:
  T:= Matrix(64,64, q);
```

$$T := \begin{bmatrix} \textit{64 x 64 Matrix} \\ \textit{Data Type: anything} \\ \textit{Storage: rectangular} \\ \textit{Order: Fortran\_order} \end{bmatrix} \quad \textbf{(1)}$$

Thus $a(n) = u\, T^n\, v$ where u and v are row and column vectors respectively with $u_i = 1$ for i corresponding to configurations with bottom row $(0, 0, 0)$, 0 otherwise, and $v_i = 1$ for i corresponding to configurations with top row $(0, 0, 0)$, 0 otherwise. The following Maple code produces these vectors.

```
> u:= Vector[row](64):
  v:= Vector(64):
  for i from 0 to 7 do u[8*i+1]:= 1; v[i+1]:= 1;
  od:
```

To check, here are the first few entries of our sequence.

```
> seq(u . T^n . v, n = 1 .. 10);
```

$$7, 33, 164, 811, 4035, 19997, 99245, 492401, 2443097, 12121712 \quad (2)$$

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$$P := t \to t^{15} - t^{14} - 12\,t^{13} - 32\,t^{12} - 32\,t^{11} + 2\,t^{10} + 40\,t^{9} + 35\,t^{8} - 6\,t^{7} - 15\,t^{6} - 15\,t^{5} + 9\,t^{4} - t^{3} + 3\,t^{2} - t \quad (3)$$

This turns out to have degree 15. Thus we will have $0 = u\,P(T)\,T^n\,v = \sum_{i=0}^{15} p_i\,a(i+n)$ where p_i is the coefficient of t^i in $P(t)$. That corresponds to a homogeneous linear recurrence of order 15, which would hold true for any u and v. It seems that with our particular u and v we have a recurrence of order only 10, corresponding to a factor of P.

```
> factor(P(t));
```

$$t\,(t^{10} - 2\,t^{9} - 10\,t^{8} - 20\,t^{7} - 17\,t^{6} + t^{5} + 9\,t^{4} + 12\,t^{3} + t^{2} + t - 1)\,(t^{4} + t^{3} - 2\,t + 1) \quad (4)$$

```
> Q:= unapply(t^10-2*t^9-10*t^8-20*t^7-17*t^6+t^5+9*t^4+12*t^3+t^2+
t-1, t);
```

$$Q := t \to t^{10} - 2\,t^{9} - 10\,t^{8} - 20\,t^{7} - 17\,t^{6} + t^{5} + 9\,t^{4} + 12\,t^{3} + t^{2} + t - 1 \quad (5)$$

The complementary factor $R(t) = \dfrac{P(t)}{Q(t)}$ has degree 5.

```
> R:= unapply(normal(P(t)/Q(t)), t);
```

$$R := t \to (t^{4} + t^{3} - 2\,t + 1)\,t \quad (6)$$

Now we want to show that $b(n) = u\,Q(T)\,T^n\,v = 0$ for all n. This will certainly satisfy the order-5 recurrence

$$\sum_{i=0}^{5} r_i\,b(i+n) = \sum_{i=0}^{5} r_i u\,Q(T)\,T^{n+i} v = u\,Q(T)\,R(T)\,T^n\,v = u\,P(T)\,T^n\,v = 0$$

where r_i are the coefficients of $R(t)$. To show all $b(n) = 0$ it suffices to show $b(0) = \ldots = b(4) = 0$.

```
> seq( u . Q(T) . T^n . v, n = 0 .. 5);
```

$$0, 0, 0, 0, 0, 0 \quad (7)$$